

Advanced Liability Structures

Triggers, Interest Rate Swaps, and Reserve Accounts

Loss protection is the single most important reason for advanced liability structures. All entities that fund transactions are worried about loss and try to anticipate and protect against its different forms. As seen from Chapter 4, nonperforming assets that have stopped generating cash and are considered delinquent or defaulted cause loss. However, there can also be structural issues such as interest rate mismatches, which need some type of protection. Advanced liability structures such as triggers, swaps, and reserve accounts are created to help prevent and mitigate these concerns.

TRIGGERS AND THEIR AFFECT ON THE LIABILITY STRUCTURE

The simplest and most cost-effective method of mitigating loss is by altering the structure of the transaction when problems arise. If a deal is performing as expected, then the liability structure is probably sufficient to ensure that all parties are repaid. However, when assets begin to default, investors worry and become very cognizant of where they stand in the priority of payments. In many structured transactions, a senior investor will have negotiated a change in the priority of payments if the deal begins to perform very poorly. The change is usually caused by a predefined test, officially known as a *trigger*, being breached. This change directs more cash to the senior investor so that the senior obligation receives principal faster.

The speed at which an investor receives principal back is often a point of confusion. While having principal returned faster is more conservative, it is not necessarily more desirable. The faster principal is returned the faster the debt obligation is paid off. A faster paying obligation will have less overall yield than a slower paying one if assets are paying as intended and debt interest and principal can be paid. Also, paying the obligation back faster changes the weighted average life and could cause a mismatch in investment tenors for an investor. This is a problem because many times investors choose which transaction to invest in with maturity and weighted average life in mind.

The opposing duality of payment speed's risk and reward makes defining and setting a trigger very difficult. If a trigger is set up too tightly, then the trigger is

breached quickly, the liability structure switches, and the investor receives principal faster than necessary. However, if a trigger is set up too loosely, then the trigger is not breached and, if there is a problem in the transaction, the investor has a higher amount of principal exposed for a longer amount of time.

A classic example of a trigger is one that is based on cumulative default rate. Imagine a set of assets that have a historical default rate of 3 percent. A structurer has decided that the transaction should have a trigger of 5 percent gross cumulative defaults, with the results of a breach being the rapid amortization of senior principal. If the assets perform as expected, historical defaults should remain around 3 percent and the senior investors get their return as expected. If defaults jump over 5 percent, then excess cash in the transaction is used to pay down the senior obligation first.

Other types of common triggers include:

- *Negative excess spread.* When excess spread, defined as the difference between the asset yield and the liability fees and interest, becomes negative this trigger is breached.
- *Delinquency.* When the delinquency rate for the assets breaches a predefined level.
- *Rolling average triggers.* These triggers take a common trigger like defaults, but use the average of a certain time period that “rolls” as time progresses, rather than just using one period as the test. This is useful because it prevents temporary spikes from changing a deal when the problem could be a single period anomaly.
- *Qualitative triggers.* There can be multiple nonquantitative triggers like missing a payment to the trust, failing to send in reports, and in general failing to meet a list of preexisting criteria.

Finally, when triggers are breached there can be many different consequences. If the trigger breached is not very severe, it could just mean trapping extra cash for a period. However, if a serious problem is occurring and a major trigger is breached the deal could then go into full rapid amortization and all cash could be redirected to senior investors. Also, triggers can be set up to cure. This means that if the trigger was breached in one period, but in the next period the metric for the trigger passes, then the state of the deal can go back to the prebreached set up. All of these nuances require a thorough understanding of triggers because they can have a very powerful impact on how a deal performs.

MODEL BUILDER 7.1: INCORPORATING TRIGGERS

1. Modeling triggers in a transaction does not necessarily mean setting up each one exactly as the documents read. Particularly in the case of qualitative triggers, this would be time consuming and most likely not worth the time since breaching any of those would be a complete guess. The only triggers that need to be modeled are ones that can be breached when the cash flow is stressed. Project

	A	B	C	D	E
27					
28		STRUCTURAL INPUTS			
29		Asset-Based Fees	2.00%		
30					
31		Capture All XS Spd	Yes		
32					
33					

FIGURE 7.1 The Capture trigger should be entered in the Structural Inputs section.

Model Builder will have four triggers to show common trigger analysis. The parameters for these triggers are located on the Inputs sheet.

Prior to entering any trigger related assumptions, a named list needs to be created on the Hidden sheet. Go to the Hidden sheet and enter the label **YesNo** in cell A25. Enter “Yes” in cell A26 and “No” in cell A27. Name the range A26:A27 **lstYesNo**.

- Next, go to the Inputs sheet and enter the label **Capture All XS Spd** in cell B31. In cell C31, create a data validation list with **lstYesNo** as the range. Name cell C31 **GlobalTrigger**. This trigger is one that will be decided by the model operator. If a “Yes” value is input in cell C31, then all excess spread in the transaction will be used to pay down senior debt. The reason for such a trigger is that a worst-case scenario is often modeled. In such a case one would assume that the assets are performing very poorly and rapid amortization triggers have already been tripped. The Inputs sheet should look like Figure 7.1.
- It is necessary to track each trigger on the Cash Flow sheet because when a trigger is breached the flow of cash in the waterfall will change. To track whether or not a trigger has been breached each period a Boolean statement (TRUE or FALSE) should be returned.

Go to the Cash Flow sheet. Enter the label **Capture Trigger** in cell AA4. In AA7 enter the following formula:

$$= \text{IF}(\text{GlobalTrigger}=\text{"Yes"}, \text{TRUE}, \text{FALSE})$$

This is a simple IF statement that directs a TRUE to be input in the cell if the range **GlobalTrigger** is set to “Yes” or FALSE if it is not. Copy this formula and paste it over the range AA7:AA366.

At this point the next logical step may seem to set up switches in the cash flow for when the trigger is breached. However, since there are three more triggers to create, it will be more efficient to set up those assumptions prior to adjusting the cash flow formulas.

- The next trigger is a more flexible version of the previous one. It allows the model operator to decide which period to begin a rapid amortization state. The reason this is useful is that as a transaction progresses under a normal state,

cash is typically released out of the transaction. Any cash released is cash that is not available for debt repayment.

A scenario that should be run, which will be discussed later in the text, is to release cash for a number of months prior to a rapid amortization event. Often triggers take a few periods to be breached, particularly in the case of default triggers where the definition of a default is three months delinquent. Such a trigger could never be breached in the first three months.

Go to the Inputs sheet and in cell B32 enter the following label, **Post-Default Trigger Month**. In cell C32 enter the value 3 for now. Name cell C32 **PostDefTriggerMo**.

5. Go to the Cash Flow sheet and in cell AB4 enter the label **Post Default Mo Trigger**. In cell AB7 enter the following formula:

```
=IF(AND(A7>=PostDefTriggerMo,PostDefTriggerMo<>0),TRUE,FALSE)
```

Deconstructing this formula reveals an AND statement that tests the current period against the value input for PostDefTriggerMo and if PostDefTriggerMo is not zero. This statement reads that if the current period in the cash flow is greater than or equal to the trigger assumption on the Inputs sheet, then the trigger has been breached, and a TRUE value should be returned. Otherwise the value is false.

Notice that when a zero is entered as the assumption the formula will return a FALSE statement. This is so there is an option to always have the trigger off. Copy the formula and paste it over the range AB7:AB366.

6. The most complicated trigger will be one that tracks defaults. If the default percentage experienced in the deal breaches a predefined level set up on the Inputs sheet, then the trigger is tripped.

Go to the Inputs sheet and enter the label **Default Trigger %** in cell B33. For now, enter 5.00% in cell C33 and name that cell **Trigger_Def**.

7. Go to the Cash Flow sheet, but go far to the right to column CP. Prior to setting up the actual trigger test a section needs to be created for tracking each period's gross cumulative default percentage. Tracking is typically done to the far right of the waterfall.

Enter the label **Cumulative Default Percentage** in cell CP4. For most transactions, the formula is going to be the current period's dollar default amount divided by the original balance. To make it cumulative the formula should add the prior period's defaulted percentage. The complete formula in cell CP7 should look as follows:

```
=N7/$L$7+CP6
```

Copy and paste this formula over the range CP7:CP366. Also, be mindful that when using an SDA curve to generate defaults, those are calculated using the current balance. Check to make sure how triggers read in every case because they can be very customized. This trigger section of the Cash Flow sheet should look like Figure 7.2.

	W	X	Y	Z	AA	AB	AC
1							
2							
3							
4		Total Cash Available for Liabilities			Capture Trigger	Post Default Mo Trigger	
5							
6							
7		681,952			TRUE	FALSE	
8		698,488			TRUE	FALSE	
9		714,900			TRUE	TRUE	
10		731,181			TRUE	TRUE	

FIGURE 7.2 The triggers on the Cash Flow sheet should start taking form.

- Still on the Cash Flow sheet go left to column AC. In cell AC4 enter the label **Default Trigger**. The formula will read very close to how the trigger is designed. When defaults exceed the amount indicated on the Inputs sheet then trip the trigger. The formula in cell AC7 should be:

$$=IF(CP7>Trigger_Def,TRUE,FALSE)$$

Copy and paste this formula over the range AC7:AC366.

- The final trigger is very simple and requires no modification to the Inputs sheet. This trigger is a custom Event of Default trigger as determined by the model operator. Occasionally the need arises for a trigger to be assumed tripped at any given point within the deal for any given amount of time. Column Z on the Cash Flow sheet will be used for this.

Label cell Z4 **Event of Default**. For now enter FALSE in cell Z7 and copy and paste this value over the range Z7:Z366. Make sure this is formatted as an input, since the model operator can change any period's value to assume a tripped trigger. At this point the Cash Flow sheet should look like Figure 7.3.

- The final part is linking the Boolean values to the cash flow structure. Three of the triggers in Project Model Builder will be used to indicate a full rapid amortization state. This means that if any of those triggers are tripped, all cash is diverted immediately to senior principal. In such a case, the subordinated tranche will be cut off from receiving funds.

Go to the Senior Principal Due column on the Cash Flow sheet (column AZ). Cell AZ7 needs to be modified to work differently when a trigger is tripped. This is going to require the use of an IF-OR combination. If the OR statement is unclear see the Toolbox section at the end of this chapter. Modify the formula in cell AZ7 as shown in bold—and make sure to enter the terminal close

	W	X	Y	Z	AA	AB	AC	AD	AE
1									
2									
3									
4		Total Cash Available for Liabilities		Event of Default	Capture Trigger	Post Default Mo Trigger	Default Trigger		
5									
6									
7		681,952		FALSE	TRUE	FALSE	FALSE		
8		698,488		FALSE	TRUE	FALSE	FALSE		
9		714,900		FALSE	TRUE	TRUE	FALSE		
10		731,181		FALSE	TRUE	TRUE	FALSE		

FIGURE 7.3 The completed trigger section of the Cash Flow sheet (with formatting).

parenthesis also shown in bold:

$$= \text{IF}(\text{OR}(\mathbf{Z7}, \mathbf{AB7}, \mathbf{AC7}), \text{MIN}(\mathbf{AX7}, \mathbf{CB6}), \text{IF}(\text{LiabPrinType1} = \text{"Sequential"}, \text{MIN}((\mathbf{N7} + \mathbf{Q7} + \mathbf{R7}), \mathbf{CB6}), \text{MIN}((\mathbf{N7} + \mathbf{Q7} + \mathbf{R7}) * \text{LiabAdvRate1}, \mathbf{CB6})))$$

The modified part has been highlighted and includes an IF and OR statement that checks to see if any of the three triggers: Event of Default (column Z), Post Default Mo Trigger (column AB), or Default Trigger (column AC) have been tripped. If any triggers have tripped, the formula calculates the Senior Principal Due as whatever amount is available at that point in the waterfall. This essentially ends the flow of cash through the waterfall at this point until the Senior Principal is completely paid off.

- The previous triggers are severe and prevent the subordinate tranche from receiving any funds. In certain cases, a trigger only accelerates the senior principal if cash remains at the end of the waterfall. The global trigger will be this type of acceleration trigger in Project Model Builder.

To make an acceleration trigger, an additional column needs to be set up at the end of the waterfall. Go to cell BY4 on the Cash Flow sheet and enter the label **Excess Applied to Sr Prin**. Cell BY7 needs a formula that returns the cash remaining if the trigger is tripped. However, the amount needs to be constrained by the balance of the senior debt. Enter the following formula in cell BY7:

$$= \text{IF}(\mathbf{AA7}, \text{MIN}(\mathbf{BW7}, \mathbf{CB6} - \mathbf{BA7}), 0)$$

This formula checks to see if the global trigger has been tripped and populates any cash remaining at the end of the waterfall. It is constrained by a MIN

	BV	BW	BX	BY	BZ	C
1						
2						
3						
4	Unpaid	Cash Remaining		Excess Applied to Sr Prin		
5						
6						
7	-	89,289		89,289		
8	-	72,438		72,438		
9	6,782	-		-		
10	7,636	-		-		

FIGURE 7.4 Excess amounts that are used for principal acceleration are calculated at the end of the waterfall.

function that takes the lesser of the amount remaining and the current balance less principal paid earlier in the waterfall. Copy this formula over the range BY7:BY366. This section should look like Figure 7.4.

- The final step is to apply the excess that was just calculated to the senior principal. Modify the formula in cell CD7 so the senior debt balance is also reduced by amounts in BY:

$$=BA7+BY7$$

If there is any excess applied to the senior principal it will reduce the balance accordingly. Copy and paste this formula over the range CD7:CD366.

SWAPS

Swaps are confusing to many people because they involve conceptual flip-flopping. Whole books are dedicated to describing what swaps are and how they work. The goal of this section to give a very brief introduction to swaps and then move on to how a basic swap can be modeled in structured transactions.

A *swap* is a financial instrument that hedges risk by swapping parties' exposure. For structured transactions an interest rate swap is the most commonly used swap. In a basic structured transaction a bank might have funded a transaction on a floating rate basis, but has structured the transaction with fixed rate assets. If the floating rate on the liabilities were to exceed the weighted average fixed rate of the assets, then the bank could take a loss.

Instead of taking such risk the bank enters into a fixed-for-floating interest rate swap. In such a case, the transaction will pay a fixed rate amount to a swap provider, while the swap provider will pay a floating rate amount to the transaction. All of

the amounts are calculated off of a notional amortization schedule, which is a base case amortization of the certificates involved in the swap.

Project Model Builder incorporates a simple fixed-for-floating interest rate swap. It should be understood that many complex features to a swap are not included in this example, such as swap dealer fees, swap termination fees, and other granularities. The purpose is to understand how a swap uses interest rates to affect cash flow in and out of a transaction.

MODEL BUILDER 7.2: INCORPORATING A BASIC INTEREST RATE SWAP

1. There are only three assumptions that need to be manipulated on the Inputs sheet: (1) whether there is a swap in a transaction, (2) the basis for the swap money coming in, and (3) the basis for the swap money going out.

On the Inputs sheet enter the following labels:

D29: **Swap Active**

D30: **Swap Rate In**

D31: **Swap Rate Out**

Cell E29 should be a data validation list with `lstYesNo` and should be named **Swap_Active**. Cells E30 and E31 should also be data validation lists, but they should use `lstInterestRates` as the range. Name these cells **Swap_In** and **Swap_Out** respectively. Also select “1-Month LIBOR” for the Swap Rate In and “Custom 1” for the Swap Rate Out. The Inputs sheet should look like Figure 7.5.

2. Next go to the Cash Flow sheet, where columns AE:AK are used for the swap calculations. Enter the following labels:

AE4: **Notional Swap Schedule**

AF4: **Swap Rate In**

AG4: **Swap Flow In**

AH4: **Swap Rate Out**

AI4 **Swap Flow Out**

AJ4: **Swap Earn/Pay**

AK4: **Cash Available**

	A	B	C	D	E
27					
28		STRUCTURAL INPUTS			
29	Asset-Based Fees	2.00%	Swap Active	Yes	
30			Swap Rate In	1-Month LIBOR	
31	Capture All XS Spd	Yes	Swap Rate Out	Custom 1	
32	Post-Default Trigger Month		3		
33	Default Trigger %	5.00%			

FIGURE 7.5 The swap inputs are included within the Structural Inputs section.

3. Column AE is where the Notional Swap Schedule is stored. This is a base case amortization of the senior certificates. For purposes of the example model, use the Notional Swap Schedule provided in Excel file *MB7-2.xls* in the Ch07 folder on the CD-ROM. Copy and paste the schedule from the CD-ROM section to the range AE7:AE366 in the model under construction. This is the assumed amortization that the swap will base cash flow on.
4. Column AF is the rate that the swap counterparty pays the transaction. In this case, the transaction needs floating rate payments so it will be a floating rate as designated on the Inputs sheet. Earlier 1-Month LIBOR was designated as the rate. Similar to the other rate formulas on the Cash Flow sheet, enter the following formula in cell AF7:

```
= IF(Swap_Active="No",0,OFFSET(Vectors!$D$6,Vectors!A7,
MATCH(Swap_In,lstInterestRates,0)))
```

While most of this formula is an OFFSET-MATCH combination that has been seen before, the beginning is an IF statement that checks to see if there is a swap in the deal or not. If not the rate will be zero, causing all calculations to be zero. Copy and paste this formula over the range AF7:AF366.

5. The dollar amount of swap flow in can be calculated with the swap rate in known. Enter the following formula in cell AG7:

```
=AE7*AF7*C7
```

This multiplies the swap rate in, by the notional schedule, and also by the day factor. Note that swaps sometimes use different day-count systems than transactions. In this example the day-count system was assumed to be the same as the transaction. Copy and paste this formula over the range AG7:AG366.

6. Prior to going to column AH, go to the Vectors sheet. Enter **4.00%** for cell I7. Copy and paste this value over the range I7:I366 so that every periods' value is 4.00 percent. This should look like Figure 7.6.
7. Go back to the Cash Flow sheet and to cell AH7. The formula here needs to return the swap rate that the transaction is paying to the swap counterparty. In this case it is a fixed rate because, on the Inputs sheet, the vector assumed is Custom 1, which in the previous step was assumed to be 4.00 percent for every period. Enter the following formula in cell AH7:

```
= IF(Swap_Active="No",0,OFFSET(Vectors!$D$6,Vectors!A7,
MATCH(Swap_Out,lstInterestRates,0)))
```

Copy and paste this formula over the range AH7:AH366.

	D	E	F	G	H	I	J	K
1								
2								
3	Interest Rates							
4		1-Month LIBOR	3-Month LIBOR	6-Month LIBOR	Prime	Custom 1	Custom 2	Custom 3
5								
6								
7		2.60%				4.00%		
8		2.81%				4.00%		
9		2.87%				4.00%		
10		2.94%				4.00%		
11		3.00%				4.00%		
12		3.06%				4.00%		

FIGURE 7.6 Make sure the Vector's sheet is updated so Cash Flow sheet calculations work.

8. The calculation for the swap flow out is identical to the swap flow in, with the exception of the referenced rate. Enter the following formula into cell AI7:

$$=AE7*AH7*C7$$

Copy and paste this formula over the range AI7:AI366.

9. To determine the net amount paid or earned from the swap, subtract the swap flow out from the swap flow in. This is done with the following formula in cell AJ7:

$$=AG7-AI7$$

Notice there is no MIN here because the value can be negative depending on the interest rate assumptions. Copy and paste this formula over the range AJ7:AJ366.

10. The swap section is completed by tracking the cash available after giving affect to swap payments. Enter the following formula in cell AK7:

$$=X7+AJ7$$

Copy and paste this formula over the range AK7:AK366. By now the swap section should look like Figure 7.7.

11. With the introduction of this advanced structure, a minor modification needs to be made to an existing formula so cash continues to flow through the waterfall. Change the formula in cell AP7 to:

$$=AK7-AN7$$

	AI	AE	AF	AG	AH	AI	AJ	AK	AI
1									
2									
3	<i>Swap</i>								
4		Notional Swap Schedule	Swap Rate In	Swap Flow In	Swap Rate Out	Swap Flow Out	Swap Earn/Pay	Cash Available	
5									
6									
7		94,171,439	2.60%	204,038	4.00%	313,905	(109,867)	572,086	
8		93,117,363	2.81%	218,197	4.00%	310,391	(92,194)	606,294	
9		92,073,966	2.87%	220,548	4.00%	306,913	(86,365)	628,535	
10		91,041,143	2.94%	222,816	4.00%	303,470	(80,655)	650,526	
11		90,018,786	3.00%	225,002	4.00%	300,063	(75,061)	672,260	
12		89,006,791	3.06%	227,108	4.00%	296,689	(69,581)	694,842	

FIGURE 7.7 The Swap section on the Cash Flow sheet is complete.

Without this change the swap calculation will have no effect on the rest of the waterfall. Make sure to copy this change down to cell AP366.

FINAL NOTES ON SWAPS

A swap can introduce complex changes to the cash flow depending on the interest rates assumed. If the rates are assumed to be extremely volatile then the swap earn/pay amounts can be very large. Keep in mind that this section has not assumed any cost for the swap. The more beneficial a swap is to a deal, the more expensive it will probably be. Swap expenses should be quoted from a swap provider who can provide up-to-the-minute market prices.

RESERVE ACCOUNTS

Reserve accounts are the most tangible and easiest form of credit enhancement to understand. They are accounts set aside exclusively for a transaction in case there are problems making payments to certain liabilities. If the liability cannot be met through normal cash flow and the deal documentation allows, a reserve account can be used to make up payment shortfall. Reserve accounts are either cash funded from the start of the transaction or they can be designed to grow by trapping excess cash in a transaction. Conversely, as deals amortize the reserve account can also amortize or stay at a fixed amount.

Issuers tend not to want cash-funded reserve accounts because the money being reserved is untouchable and not earning a high return. It is important that the

money can only be accessed for certain obligations; otherwise there is little value in assuming the reserve amount.

Another important feature of reserve accounts is that they are typically reimbursed if there is enough cash in the transaction. A minimum reserve amount is often required and when the reserve balance goes below the minimum, reimbursements are necessary. This is important for the methodology being implemented in Project Model Builder because the placement of the reserve account calculations depends on where reimbursements are written into the priority of payments.

MODEL BUILDER 7.3: INCORPORATING A CASH-FUNDED RESERVE ACCOUNT

1. A cash-funded reserve account is assumed in Project Model Builder. This type of reserve account has cash funded by the asset issuer, which is typically a percent of the assets. In the deal documentation there will be language that designates what liabilities the reserve account covers. Project Model Builder assumes that only the senior liabilities have access to the reserve account. Typically fees and top-level items on the waterfall have access to the reserve account, but modeling this is not necessary because very few entities would do a deal that is risky enough where the top of the waterfall has the possibility of drawing from a reserve account.

To start modeling the reserve account go to the Inputs sheet and enter the following label in cell I23, **Reserve Active**. Make cells I24 and I25 data validation lists using `IstYesNo` as the range. Name cell I24 **LiabReserveOnOff1** and I25 **LiabReserveOnOff2**. Enter the label **Reserve Account %** in B30 and the value **1.00%** in C30. Name cell C30 **RsrvPercent**. The Inputs sheet should look like Figure 7.8.

2. Go to the Cash Flow sheet and enter the following labels:
BG4: Reserve Account Minimum
BH4: Reserve Account Beginning Balance
BI4: Withdrawals

	A	B	C	D	E	F	G	H	I	J
21										
22										
23										
24										
25										
26										
27										
28										
29										
30										
31										
32										
33										

LIABILITY INPUTS								
Debt Description	Advance Rate	Liability Interest Type	Floating Rate Curve	Fixed Rate	Loan Margin	Fees	Reserve Active	Prin Allocation Type
Senior Debt 1	95.00%	Floating	1-Month LIBOR		1.00%	0.50%	Yes	Pro-Rate
Sub Loan	5.00%	Floating	Custom 1		0.00%	0.00%	No	Pro-Rate

STRUCTURAL INPUTS			
Asset Based Fees	2.00%	Srvap Active	Yes
Reserve Account %	1.00%	Srvap Rate In	1-Month LIBOR
Capture All X3 Spd	Yes	Srvap Rate Out	Custom 1
Post-Default Trigger Month			3
Default Trigger %	5.00%		

FIGURE 7.8 The Reserve Account additions are in multiple sections of the Inputs sheet.

BJ4: Reimbursements**BK4: Reserve Account Ending Balance****BL4: Cash Remaining**

3. The reserve account minimum is the first concept in this section. It is the amount that should be maintained in the reserve account each period. If the deal has an amortizing reserve, this amount will decrease as the asset pool balance decreases. However, in Project Model Builder the reserve is a fixed amount. Enter the following formula in cell BG7:

$$=AssetCurBal1 * RsrvPercent$$

This formula multiplies the reserve percent from the Inputs sheet by the asset pool's beginning balance. Copy and paste this formula over the range BG7:BG366.

4. To calculate a reserve section the beginning of period and end of period balance should be split into separate columns. In this example, the beginning of period reserve balance is always the end of period balance from the prior period. Enter the following formula in cell BH7:

$$BK6$$

Copy and paste the formula over the range BH7:BH366.

5. In order to populate values while working with the reserve account section, skip over to cell BK6. Enter the following formula:

$$=IF(A6=0,AssetCurBal1 * RsrvPercent,BH6-BI6+BJ6)$$

This formula begins by checking to see if the current period is the start of the transaction (period 0). If it is then the reserve account is assumed to start with a funded balance of the reserve percent multiplied by the asset pool balance. Otherwise the ending balance will be the beginning balance minus withdrawals plus reimbursements.

6. The next focus of this section is withdrawals from the reserve account. Withdrawals should only be made if the deal documentation allows. In this model assume that only senior interest and principal are covered by the reserve account. Go to cell AV4 and enter the label **Unpaid Covered by Reserve**. Next go to cell AW4 and enter the label **Unpaid**. In cell AV7 enter:

$$=IF(LiabReserveOnOff1="No",0,MIN(AU7,BH7))$$

This formula first checks to see if the reserve is active. If it is not active then there is no coverage of unpaid amounts. If the reserve is active then the lesser of “What You Have and What You Need” is applied. The needed part is the unpaid interest that is calculated in cell AU7, while the “have” part is the beginning balance of the reserve account for the period in cell BH7. With this

set up the amount covered by the reserve will never be more than the reserve account balance. Copy and paste this formula over the range AV7:AV366.

7. If there is not enough cash in the reserve account the unpaid amount needs to be carried over. Label cell AW4 **Unpaid** and enter the following formula in AW7:

$$=AU7-AV7$$

This subtracts the amount covered by the reserve from the unpaid amount. Copy and paste this formula over the range AW7:AW366.

8. A similar process now needs to take place for the senior principal. Label cell BC4 **Unpaid Covered By Reserve**. In BC7 enter the following formula:

$$=IF(LiabReserveOnOff1="No",0,MIN(BB7,BH7-AV7))$$

The major difference in this formula is that the amount used from the reserve (cell AV7) is subtracted from the reserve account balance (cell BH7). This is a logical assumption if one assumes that the cash flow waterfall has a time element from left to right. First, any unpaid interest would be paid from the reserve account and then unpaid principal would be covered only if there is money available left in the reserve account. Copy and paste this formula over the range BC7:BC366.

9. The formula for tracking the carried over unpaid amounts also needs to be created. Label cell BD4 **Unpaid** and in BD7 enter:

$$=BB7-BC7$$

Copy and paste this formula over the range BD7:BD366.

10. Go back over to the reserve account section to cell BI7. The amount withdrawn is the sum of the fields that calculate amounts covered by the reserve account. Enter the following formula in cell BI7:

$$=AV7+BC7$$

Copy and paste this formula over the range BI7:BI366.

11. Calculating reimbursements is the next and perhaps the most complicated part of modeling reserve accounts. The amount to be reimbursed should always be the reserve minimum less the reserve account beginning of period balance. Reimbursements are not necessary when the debt that is being covered by the reserve is paid off and the reserve account is liquidated, or when the reserve balance is at or above the reserve minimum. Enter the following formula in cell BJ7 to accomplish all of this:

$$=IF(CB6<1,0,MAX(MIN(BG7-BH7,BE7),0))$$

This formula first checks the beginning balance of the debt being covered by the reserve (in this case the senior debt) and makes reimbursements zero if the debt is paid off. If the debt still exists then the formula performs a lesser of “What You Have and What You Need.” In this calculation, the reserve account balance subtracted from the reserve account minimum is needed, while the cash remaining in cell BE7 is available.

However, occasionally the reserve account balance might be higher than the minimum. If that is the case then this formula will produce a negative result. The MAX formula insures that there are no negative reimbursements in such cases. Copy and paste the formula in BJ7 over the range BJ7:BJ366.

12. The reserve account section is completed by creating a cash remaining calculation by entering the following formula in cell BL7:

$$=BE7-BJ7$$

This formula is important because it shows that the reimbursements are removed from the cash available at this point in the cash flow waterfall. If reimbursements were anywhere else in the documentation, then this reference would have to be moved accordingly. Copy and paste this formula over the range BL7:BL366. The entire reserve account section should look like Figure 7.9.

13. A few modifications need to be made to existing formulas to make the reserve account work correctly. First, cell BR7 needs to be modified to:

$$=BL7-BP7$$

This will use the cash remaining from the reserve account section rather than skipping over all of the reserve account calculations. Make sure to copy the formula down to BR366.

	BI	BG	BH	BI	BJ	BK	BL	BI	
1									
2									
3	Reserve Account								
4		Reserve Account Minimum	Reserve Account Beginning Balance	Withdrawals	Reimbursements	Reserve Account Ending Balance	Cash Remaining		
5									
6						1,000,000			
7		1,000,000	1,000,000	15,507	-	984,493	-		
8		1,000,000	984,493	13,865	-	970,608	-		
9		1,000,000	970,608	-	-	970,608	-		
10		1,000,000	970,608	-	-	970,608	-		
11		1,000,000	970,608	-	-	970,608	-		

FIGURE 7.9 The Reserve Account section on the Cash Flow sheet.

14. Also, the senior interest and principal aggregations in the balance section need to include amounts covered by the reserve account. Change CC7 to:

$$=AT7+AV7$$

And CD7 to:

$$=BA7+BC7+BY7$$

Make sure to copy the changes down to row 366 for each column.

15. The final step to finish the operating part of the cash flow waterfall is tracking excess cash. In cell BZ4 enter the label **Excess Released**. In cell BZ7, enter the following formula:

$$=BW7-BY7$$

This formula subtracts any excess cash that was used to pay down principal from the cash remaining after sub loan principal is paid (essentially the end of the waterfall). The amounts in this column will be released from the transaction to whoever holds the rights to the excess. Copy and paste this formula over the range BZ7:BZ366.

16. One final modification needs to be made to the Principal Due calculation in column AZ. Modify AZ7 with the following shown in bold:

$$= \text{IF}(\text{OR}(Z7, AB7, AC7), \text{MIN}(AX7, AX7+BH7-AV7, CB6), \\ \text{IF}(\text{LiabPrinType1}="Sequential", \text{MIN}((N7+Q7+R7), CB6), \\ \text{MIN}((N7+Q7+R7)*\text{LiabAdvRate1}, CB6)))$$

What this slight change does is require the cash reserve to be used in the case of a trigger breach. Most transactions will use the cash reserve in such a manner, but each deal's documentation should be checked to see how the components operate. Copy and paste cell AZ over the range AZ7:AZ366.

CONCLUSION OF THE CASH FLOW WATERFALL

The cash flow waterfall is now completely operational. However, it should be checked and formatted. All cash should flow through from left to right and down. Make sure to check the model under construction with the completed model so that the calculations are the same. This can be done by taking the sum of many of the individual columns in row 5 and checking to see if the sums are the same as those in row 5 of the completed model.

Also notice that in row 3 of the completed model there are names for the different sections. While these have no calculation value, they are helpful for jumping between sections in the waterfall by using CTRL + arrow keyboard commands.

Finally, as mentioned earlier, the color system used is useful for grouping similar concepts. Using the same color scheme for concepts in every model built allows a model operator to quickly identify sections of the model. Also, the grey separation lines are useful to break up concepts and create a “modular” type system. The breaks are particularly useful when sections of the waterfall need to be added or removed.

The remainder of the book focuses on additions to the model that ensure it is operating properly, efficient methods and tools for extracting and manipulating data, analysis of the data produced by the model, and in general understanding the model that has been created.

TOOLBOX

AND and OR

AND and OR are two important functions used in this chapter. Understanding the difference between them is simple but extremely important. An AND function returns a TRUE value if all the conditions in the formula are true. If just one of the conditions set up in the AND formula are false, then the return value is false. An OR function returns a TRUE value if any of the conditions in the formula are true. It does not matter how many conditions are false.

These two functions are very effective at translating written tests into a computer model. They are perfect for sections in a term sheet, where certain conditions must be met for an event to take place.